Common and Unique Latent Transition Analysis (CULTA) as a Way to Examine the Trait-State Dynamics of Alcohol Intoxication (Supplementary Materials)

Ivan Jacob Agaloos Pesigan[1], Michael A. Russell[1, 2], and Sy-Miin Chow[3]

[1]Edna Bennett Pierce Prevention Research Center, The Pennsylvania State University

[2]Department of Biobehavioral Health, The Pennsylvania State University

[3]Department of Human Development and Family Studies, The Pennsylvania State University

Author Note

Ivan Jacob Agaloos Pesigan https://orcid.org/0000-0003-4818-8420; Michael A. Russell https://orcid.org/0000-0002-3956-604X; Sy-Miin Chow https://orcid.org/0000-0003-1938-027X.

This research was made possible by the Prevention and Methodology Training Program (PAMT) funded by a T32 training grant (T32 DA017629 MPIs: J. Maggs & S. Lanza) from the National Institute on Drug Abuse (NIDA).

Add your grants...

Computations for this research were performed on the Pennsylvania State University's Institute for Computational and Data Sciences' Roar supercomputer using SLURM for job scheduling (Yoo et al., 2003), GNU Parallel to run the simulations in parallel (Tange, 2021), and Apptainer to ensure a reproducible software stack (Kurtzer et al., 2017, 2021).

Correspondence concerning this article should be addressed to Ivan Jacob Agaloos Pesigan, Edna Bennett Pierce Prevention Research Center, College of Health and Human Development, The Pennsylvania State University, 320 Biobehavioral Health Building, University Park, PA 16802 or by email (ijapesigan@psu.edu).

Common and Unique Latent Transition Analysis (CULTA) as a Way to Examine the Trait-State Dynamics of Alcohol Intoxication (Supplementary Materials)

Contents

Links

Research Compendium

The data and materials for this study are available on OSF (https://osf.io/gtdmr) and GitHub (https://github.com/jeksterslab/manCULTA, https://jeksterslab.github.io/manCULTA/index.html).

Data Simulation and Model Fitting

https://jeksterslab.github.io/manCULTA/articles/sim-culta-2-profiles.html

Comparison of Misspecified and Correctly Specified Models

- One-Profile CULTA vs. Two-Profile CULTA:
  https://jeksterslab.github.io/manCULTA/articles/sim-culta-1-profile.html

- Two-Profile LTA vs. Two-Profile CULTA:
  https://jeksterslab.github.io/manCULTA/articles/sim-lta-2-profiles.html

- Two-Profile RI-LTA vs. Two-Profile CULTA:
  https://jeksterslab.github.io/manCULTA/articles/sim-ri-lta-2-profiles.html

Generating Mplus Input Files

https://jeksterslab.github.io/manCULTA/articles/sim-input.html

Containers for Reproducibility

https://jeksterslab.github.io/manCULTA/articles/containers.html

Mplus Input File for the Empirical Data Analysis

Final Model

```
TITLE:
  2-Profile CULTA with Covariate (Final);

DATA:
  FILE = __DATA__;

VARIABLE:
  NAMES =
    id x
    y1t0 y2t0 y3t0 y4t0 y1t1 y2t1 y3t1 y4t1
    y1t2 y2t2 y3t2 y4t2 y1t3 y2t3 y3t3 y4t3
    y1t4 y2t4 y3t4 y4t4 y1t5 y2t5 y3t5 y4t5
  ;
  USEVARIABLES =
    x
    y1t0 y2t0 y3t0 y4t0 y1t1 y2t1 y3t1 y4t1
    y1t2 y2t2 y3t2 y4t2 y1t3 y2t3 y3t3 y4t3
    y1t4 y2t4 y3t4 y4t4 y1t5 y2t5 y3t5 y4t5
  ;
  IDVARIABLE = id;
  CLASSES = c0(2) c1(2) c2(2) c3(2) c4(2) c5(2);
  MISSING = .;

DEFINE:
  STANDARDIZE
    y1t0 y2t0 y3t0 y4t0 y1t1 y2t1 y3t1 y4t1
    y1t2 y2t2 y3t2 y4t2 y1t3 y2t3 y3t3 y4t3
    y1t4 y2t4 y3t4 y4t4 y1t5 y2t5 y3t5 y4t5
  ;
```

```
ANALYSIS:
  TYPE = MIXTURE;
  STARTS = 200 100;
  STSCALE = 2;
  STITERATIONS = 200;
  PROCESS = __CORES__;
  MODEL = NOCOV;

MODEL:
  %OVERALL%
    ! unique traits -----------------------------------------------------

    !! factor loadings
    !!! k = 3
    u3 BY y3t0@1;
    u3 BY y3t1@1;
    u3 BY y3t2@1;
    u3 BY y3t3@1;
    u3 BY y3t4@1;
    u3 BY y3t5@1;
    !!! k = 4
    u4 BY y4t0@1;
    u4 BY y4t1@1;
    u4 BY y4t2@1;
    u4 BY y4t3@1;
    u4 BY y4t4@1;
    u4 BY y4t5@1;

    !! latent means
    [ u3@0 ];
    [ u4@0 ];
```

```
!! latent variances
u3 (psip3);
u4 (psip4);

! common states ----------------------------------------------------

!! factor loadings
!!! t = 0
s0 BY y1t0@1;
s0 BY y2t0 (lambdas2);
s0 BY y3t0 (lambdas3);
s0 BY y4t0 (lambdas4);
!!! t = 1
s1 BY y1t1@1;
s1 BY y2t1 (lambdas2);
s1 BY y3t1 (lambdas3);
s1 BY y4t1 (lambdas4);
!!! t = 2
s2 BY y1t2@1;
s2 BY y2t2 (lambdas2);
s2 BY y3t2 (lambdas3);
s2 BY y4t2 (lambdas4);
!!! t = 3
s3 BY y1t3@1;
s3 BY y2t3 (lambdas2);
s3 BY y3t3 (lambdas3);
s3 BY y4t3 (lambdas4);
!!! t = 4
s4 BY y1t4@1;
s4 BY y2t4 (lambdas2);
s4 BY y3t4 (lambdas3);
s4 BY y4t4 (lambdas4);
```

```
!!! t = 5
s5 BY y1t5@1;
s5 BY y2t5 (lambdas2);
s5 BY y3t5 (lambdas3);
s5 BY y4t5 (lambdas4);

!! latent means
[ s0@0 ];
[ s1@0 ];
[ s2@0 ];
[ s3@0 ];
[ s4@0 ];
[ s5@0 ];

!! latent variance of s0
s0 (psis0);

!! variance of the process noise
s1 (psis);
s2 (psis);
s3 (psis);
s4 (psis);
s5 (psis);

! unique states ----------------------------------------------------

!! variances
!!! t = 0
y1t0 (theta11);
y2t0 (theta22);
y3t0 (theta33);
y4t0 (theta44);
```

```
!!! t = 1
y1t1 (theta11);
y2t1 (theta22);
y3t1 (theta33);
y4t1 (theta44);
!!! t = 2
y1t2 (theta11);
y2t2 (theta22);
y3t2 (theta33);
y4t2 (theta44);
!!! t = 3
y1t3 (theta11);
y2t3 (theta22);
y3t3 (theta33);
y4t3 (theta44);
!!! t = 4
y1t4 (theta11);
y2t4 (theta22);
y3t4 (theta33);
y4t4 (theta44);
!!! t = 5
y1t5 (theta11);
y2t5 (theta22);
y3t5 (theta33);
y4t5 (theta44);

! constrained intercepts -------------------------------------------

!! t = 0
[ y1t0@0 ];
[ y2t0@0 ];
[ y3t0@0 ];
```

```
[ y4t0@0 ];
!! t = 1
[ y1t1@0 ];
[ y2t1@0 ];
[ y3t1@0 ];
[ y4t1@0 ];
!! t = 2
[ y1t2@0 ];
[ y2t2@0 ];
[ y3t2@0 ];
[ y4t2@0 ];
!! t = 3
[ y1t3@0 ];
[ y2t3@0 ];
[ y3t3@0 ];
[ y4t3@0 ];
!! t = 4
[ y1t4@0 ];
[ y2t4@0 ];
[ y3t4@0 ];
[ y4t4@0 ];
!! t = 5
[ y1t5@0 ];
[ y2t5@0 ];
[ y3t5@0 ];
[ y4t5@0 ];

! lta ---------------------------------------------------------------

!! initial profile membership
[ c0#1 ] (nu0);
c0#1 ON x (kappa0);
```

```
    !! profile transitions
    [ c1#1 ] (alpha0);
    [ c2#1 ] (alpha0);
    [ c3#1 ] (alpha0);
    [ c4#1 ] (alpha0);
    [ c5#1 ] (alpha0);
    c1#1 ON c0#1 (beta00);
    c2#1 ON c1#1 (beta00);
    c3#1 ON c2#1 (beta00);
    c4#1 ON c3#1 (beta00);
    c5#1 ON c4#1 (beta00);

MODEL c0:
  %c0#1%
    ! profile specific means
    [ y1t0 ] (mu10);
    [ y2t0 ] (mu20);
    [ y3t0 ] (mu30);
    [ y4t0 ] (mu40);

    ! covariate
    c1 ON x (gamma00);

  %c0#2%
    ! profile specific means
    [ y1t0 ] (mu11);
    [ y2t0 ] (mu21);
    [ y3t0 ] (mu31);
    [ y4t0 ] (mu41);

    ! covariate
    c1 ON x (gamma10);
```

```
MODEL c1:
  %c1#1%
    ! profile specific means
    [ y1t1 ] (mu10);
    [ y2t1 ] (mu20);
    [ y3t1 ] (mu30);
    [ y4t1 ] (mu40);

    ! covariate
    c2 ON x (gamma00);

    ! inertia
    s1 ON s0@0 (phi0);

  %c1#2%
    ! profile specific means
    [ y1t1 ] (mu11);
    [ y2t1 ] (mu21);
    [ y3t1 ] (mu31);
    [ y4t1 ] (mu41);

    ! covariate
    c2 ON x (gamma10);

    ! inertia
    s1 ON s0 (phi1);

MODEL c2:
  %c2#1%
    ! profile specific means
    [ y1t2 ] (mu10);
    [ y2t2 ] (mu20);
```

```
    [ y3t2 ] (mu30);
    [ y4t2 ] (mu40);

    ! covariate
    c3 ON x (gamma00);

    ! inertia
    s2 ON s1@0 (phi0);

  %c2#2%
    ! profile specific means
    [ y1t2 ] (mu11);
    [ y2t2 ] (mu21);
    [ y3t2 ] (mu31);
    [ y4t2 ] (mu41);

    ! covariate
    c3 ON x (gamma10);

    ! inertia
    s2 ON s1 (phi1);

MODEL c3:
  %c3#1%
    ! profile specific means
    [ y1t3 ] (mu10);
    [ y2t3 ] (mu20);
    [ y3t3 ] (mu30);
    [ y4t3 ] (mu40);

    ! covariate
    c4 ON x (gamma00);
```

```
    ! inertia
    s3 ON s2@0 (phi0);

  %c3#2%
    ! profile specific means
    [ y1t3 ] (mu11);
    [ y2t3 ] (mu21);
    [ y3t3 ] (mu31);
    [ y4t3 ] (mu41);

    ! covariate
    c4 ON x (gamma10);

    ! inertia
    s3 ON s2 (phi1);

MODEL c4:
  %c4#1%
    ! profile specific means
    [ y1t4 ] (mu10);
    [ y2t4 ] (mu20);
    [ y3t4 ] (mu30);
    [ y4t4 ] (mu40);

    ! covariate
    c5 ON x (gamma00);

    ! inertia
    s4 ON s3@0 (phi0);

  %c4#2%
    ! profile specific means
```

```
    [ y1t4 ] (mu11);
    [ y2t4 ] (mu21);
    [ y3t4 ] (mu31);
    [ y4t4 ] (mu41);

    ! covariate
    c5 ON x (gamma10);

    ! inertia
    s4 ON s3 (phi1);

MODEL c5:
  %c5#1%
    ! profile specific means
    [ y1t5 ] (mu10);
    [ y2t5 ] (mu20);
    [ y3t5 ] (mu30);
    [ y4t5 ] (mu40);

    ! inertia
    s5 ON s4@0 (phi0);

  %c5#2%
    ! profile specific means
    [ y1t5 ] (mu11);
    [ y2t5 ] (mu21);
    [ y3t5 ] (mu31);
    [ y4t5 ] (mu41);

    ! inertia
    s5 ON s4 (phi1);
```

```
MODEL CONSTRAINT:
  ! means for the first profile are higher than the second
  mu10 > mu11;
  mu20 > mu21;
  mu30 > mu31;
  mu40 > mu41;

  ! make sure variances are greater than zero
  psip3 > 0;
  psip4 > 0;
  psis0 > 0;
  psis > 0;
  theta11 > 0;
  theta22 > 0;
  theta33 > 0;
  theta44 > 0;

OUTPUT:
  TECH1 TECH3 TECH4 TECH7 TECH8 TECH12 TECH15 ENTROPY;

SAVEDATA:
  ESTIMATES = __ESTIMATES__;
  RESULTS = __RESULTS__;
  TECH3 = __TECH3__;
  TECH4 = __TECH4__;
  FILE = __CPROB__;
  SAVE = CPROBABILITIES;
```

Monte Carlo Simulation to Evaluate the CULTA Model

The Monte Carlo simulation evaluated the recovery of parameters from the Common and Unique Latent Transition Analysis (CULTA) model under three levels of latent profile separation: high (HI), moderate (MO), and low (LO) and three samples sizes: $N = 100$, $N = 200$, and $N = 300$. Figure 1 presents the indicator means for the two-profile solution under each separation condition.

The CULTA model was compared to two alternative models: standard Latent Transition Analysis (LTA) and Random-Intercept Latent Transition Analysis (RILTA). Model performance was assessed using several evaluation metrics. Figure 2 shows information criteria (e.g., AIC, BIC), while Figure 3 presents entropy values as indicators of classification quality.

Parameter recovery was examined in terms of bias (Figure 4), root mean square error (RMSE; Figure 5), coverage probability of 95% confidence intervals (Figure 6), and statistical power (Figure 7). These evaluations were conducted across 18 key parameters common to all models listed in Table 1, including variances, intercepts, covariate effects, transition parameters, and profile-specific means.

The results highlight the impact of profile separation on estimation accuracy, with reduced performance observed under low-separation conditions. Across most conditions, the CULTA model showed advantages in balancing model fit, classification accuracy, and parameter recovery relative to LTA and RILTA.

Table 1
Parameters

No.	Symbol	Description
1	θ_{11}	Unique state variance for peak (item 1).
2	θ_{22}	Unique state variance for rise (item 2).
3	θ_{33}	Unique state variance for fall (item 3).
4	θ_{44}	Unique state variance for duration (item 4).
5	ν_0	Intercept for initial log-odds of profile 0 (vs. profile 1) when $X = 0$.
6	κ_0	Covariate effect on initial profile membership; higher X increases odds of profile 0.
7	α_0	Baseline log-odds of being in profile 0 across days.
8	β_{00}	Increased odds of staying in profile 0 if previously in that profile; reflects persistence.
9	γ_{00}	Covariate effect on staying in profile 0; higher X increases persistence.
10	γ_{10}	Covariate effect on switching from state to profile 0; higher X increases transition odds.
11	μ_{10}	Profile specific mean for profile 0 and peak (item 1).
12	μ_{20}	Profile specific mean for profile 0 and rise (item 2).
13	μ_{30}	Profile specific mean for profile 0 and fall (item 3).
14	μ_{40}	Profile specific mean for profile 0 and duration (item 4).
15	μ_{11}	Profile specific mean for profile 1 and peak (item 1).
16	μ_{21}	Profile specific mean for profile 1 and rise (item 2).
17	μ_{31}	Profile specific mean for profile 1 and fall (item 3).
18	μ_{41}	Profile specific mean for profile 1 and duration (item 4).

Figure 1

Latent Profile Indicator Means

(a) *High Separation (HI)*



(b) *Moderate Separation (MO)*



(c) *Low Separation (LO)*



Figure 2

Information Criteria



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation. AIC = Aikaike Information Criteria. BIC = Bayesian Information Criteria. aBIC = sample size adjusted BIC.

Figure 3
Entropy



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation.

Figure 4

Bias



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation. See Table 1 for the parameters in the x-axis.

Figure 5

Root Mean Square Error



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation. See Table 1 for the parameters in the x-axis.

Figure 6
Coverage Probability



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation. See Table 1 for the parameters in the x-axis.

Figure 7

Statistical Power



Note: CULTA = Common and Unique Latent Transition Analysis. LTA = Latent Transition Analysis. RILTA = Random-Intercept Latent Transition Analysis. LO = Low separation. MO = Moderate separation. HI = High separation. See Table 1 for the parameters in the x-axis.